Avant!

Chapter 2

Getting Started

The examples in this chapter show you how to run Star-Hspice to perform some simple analyses.

This chapter includes the following examples:

- AC Analysis of an RC Network
- Transient Analysis of an RC Network
- Transient Analysis of an Inverter

AC Analysis of an RC Network

Figure 2-1: shows a simple RC network with a DC and AC source applied. The circuit consists of two resistors, R1 and R2, capacitor C1, and the source V1. Node 1 is the connection between the source positive terminal and R1. Node 2 is where R1, R2, and C1 are connected. Star-Hspice ground is always node 0.



Figure 2-1: – RC Network Circuit

The Star-Hspice netlist for the RC network circuit is:

```
A SIMPLE AC RUN
.OPTIONS LIST NODE POST
.OP
.AC DEC 10 1K 1MEG
.PRINT AC V(1) V(2) I(R2) I(C1)
V1 1 0 10 AC 1
R1 1 2 1K
R2 2 0 1K
C1 2 0 .001U
.END
```

Follow the procedure below to perform an AC analysis for the RC network circuit.

1. Type the above netlist into a file named *quickAC.sp*.
2. Run a Star-Hspice analysis by typing

   ```
   hspice quickAC.sp > quickAC.lis
   ```

   When the run finishes Star-Hspice displays

   ```
   >info:     ***** hspice job concluded
   ```

   followed by a line that shows the amount of real time, user time, and system time needed for the analysis.

   The following new files are present in your run directory:
   quickAC.ac0
   quickAC.ic
   quickAC.lis
   quickAC.st0.
3. Use an editor to view the *.lis* and *.st0* files to examine the simulation results and status.
4. Run AvanWaves and open the *.sp* file. Select the *quickAC.ac0* file from the Results Browser window to view the waveform. Display the voltage at node 2, using a log scale on the x-axis.

Figure 2-2: shows the waveform that was produced by sweeping the response of node 2 as the frequency of the input was varied from 1 kHz to 1 MHz.



Figure 2-2: RC Network Node 2 Frequency Response

The file *quickAC.lis* displays the input netlist, details about the elements and topology, operating point information, and the table of requested data as the input is swept from 1 kHz to 1 MHz. The files *quickAC.ic* and *quickAC.st0* contain information about the DC operating point conditions and the Star-Hspice run status, respectively. The operating point conditions can be used for subsequent simulation runs using the .LOAD statement.

Transient Analysis of an RC Network

As a second example, run a transient analysis using the same RC network as in Figure 2-1:, but adding a pulse source to the DC and AC sources.

1. Type the following equivalent Star-Hspice netlist into a file named *quickTRAN.sp*.

```
A SIMPLE TRANSIENT RUN
.OPTIONS LIST NODE POST
.OP
.TRAN 10N 2U
.PRINT TRAN V(1) V(2) I(R2) I(C1)
V1 1 0 10 AC 1 PULSE 0 5 10N 20N 20N 500N 2U
R1 1 2 1K
R2 2 0 1K
C1 2 0 .001U
.END
```

   Note that the V1 source specification has added a pulse source. The syntax for pulsbe sources and other types of sources is described in Chapter 5, “Using Sources and Stimuli”.

2. Type the following to run Star-Hspice.

```
hspice quickTRAN.sp > quickTRAN.lis
```

3. Use an editor to view the *.lis* and *.st0* files to examine the simulation results and status.

4. Run AvanWaves and open the *.sp* file. Select the *quickTRAN.tr0* file from the Results Browser window to view the waveform. Display the voltage at nodes 1 and 2 on the x-axis.

The waveforms are shown in Figure 2-3:ob.



Figure 2-3: Voltages at RC Network Circuit Node 1 and Node 2

Transient Analysis of an Inverter

As a final example, analyze the behavior of the simple MOS inverter shown in Figure 2-4:.



Figure 2-4: – MOS Inverter Circuit

1. Type the following netlist data into a file named *quickINV.sp*.

```
Inverter Circuit
.OPTIONS LIST NODE POST
.TRAN 200P 20N
.PRINT TRAN V(IN) V(OUT)
M1 OUT IN VCC VCC PCH L=1U W=20U
M2 OUT IN 0 0 NCH L=1U W=20U
VCC VCC 0 5
VIN IN 0 0 PULSE .2 4.8 2N 1N 1N 5N 20N
CLOAD OUT 0 .75P
.MODEL PCH PMOS LEVEL=1
.MODEL NCH NMOS LEVEL=1
.END
```

2. Type the following to run Star-Hspice.

```
hspice quickINV.sp > quickINV.lis
```

Use AvanWaves to examine the voltage waveforms at the inverter IN and OUT nodes. The waveforms are shown in Figure 2-5:.



Figure 2-5: – Voltage at MOS Inverter Node 1 and Node 2